

01

```
PREFIX i: <http://example.cz/is#>
SELECT ?number ?first ?last
FROM <http://example.cz/students/>
WHERE {
  ?s i:number ?number ;
     i:name [ i:first ?first ; i:last ?last ] .
}

... SELECT $number $first $last
... WHERE {
  $s i:number $number ;
     i:name [ i:first $first ; i:last $last ] .
}

... WHERE {
  ?s i:number ?number ;
     i:name ?n .
  ?n i:first ?first ; i:last ?last .
}

... WHERE {
  ?s i:number ?number .
  ?s i:name ?n .
  ?n i:first ?first .
  ?n i:last ?last .
}
```

02

```
PREFIX i: <http://example.cz/is#>
SELECT ?code ?title
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code ; i:title ?title .
  FILTER ( regex(?code, "^NPRG") )
}
ORDER BY ?title

... FILTER regex(?code, "^NPRG") ...

... ORDER BY ASC(?title)
```

03

```
PREFIX i: <http://example.cz/is#>
SELECT ?number ?email
FROM <http://example.cz/students/>
WHERE {
  ?s i:number ?number .
  OPTIONAL { ?s i:email ?email . }
}
```

04

```
PREFIX i: <http://example.cz/is#>
SELECT ?number ?email ?web
FROM <http://example.cz/students/>
WHERE {
  ?s i:number ?number
  OPTIONAL { ?s i:email ?email }
  OPTIONAL { ?s i:web ?web }
}
```

Incorrect:

```
... WHERE {
  ?s i:number ?number
  OPTIONAL {
    ?s i:email ?email
    OPTIONAL { ?s i:web ?web }
  }
}
```

Incorrect:

```
... WHERE {
  ?s i:number ?number
  OPTIONAL { ?s i:email ?email ; i:web ?web }
}
```

05

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT DISTINCT ?c ?code
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code .
  ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
  FILTER ( (?day = "MON") || (?day = "FRI") )
}

... WHERE {
  ?c i:code ?code .
  ?e i:course ?c ; i:day ?day ; i:term ?term .
  FILTER ( (?day = "MON") || (?day = "FRI") )
  FILTER ( ?term = t:term1112W )
}

... WHERE {
  {
    ?c i:code ?code .
    ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
    FILTER ( ?day = "MON" )
  } UNION {
    ?c i:code ?code .
    ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
    FILTER ( ?day = "FRI" )
  }
}
```

06

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT DISTINCT ?c ?code
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code .
  OPTIONAL {
    ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
    FILTER ( (?day = "MON") || (?day = "FRI") )
  }
  FILTER ( ! bound(?e) )
}

... WHERE {
  ?c i:code ?code .
  OPTIONAL {
    ?e1 i:course ?c ; i:day "MON" ; i:term t:term1112W
  }
  OPTIONAL {
    ?e2 i:course ?c ; i:day "FRI" ; i:term t:term1112W
  }
  FILTER ( ! bound(?e1) && ! bound(?e2) )
}
```

Incorrect:

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT DISTINCT ?c ?code
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code .
  ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
  FILTER ( (?day != "MON") && (?day != "FRI") )
}
```

07:A

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT DISTINCT ?c ?code
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code .
  FILTER NOT EXISTS {
    ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
    FILTER ( (?day = "MON") || (?day = "FRI") )
  }
}
```

07:B

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT DISTINCT ?c ?code
FROM <http://example.cz/teaching/>
WHERE {
  ?c i:code ?code .
  MINUS {
    ?e i:course ?c ; i:day ?day ; i:term t:term1112W .
    FILTER ( (?day = "MON") || (?day = "FRI") )
  }
}
```

08

```
PREFIX i: <http://example.cz/is#>
PREFIX t: <http://example.cz/teaching/>
SELECT ?first ?last (AVG(?r) AS ?average)
FROM <http://example.cz/students/>
WHERE {
    ?s i:name [ i:first ?first ; i:last ?last ] ;
    i:enroll [ i:course ?c ; i:result ?r ; i:term t:term1112W ] .
}
GROUP BY ?s ?first ?last
HAVING (COUNT(?c) >= 10)
```